

Programação com a linguagem Céu

2 players - continuação

Anny Caroline

`annycarolinegnr@gmail.com`

Francisco Sant'Anna

`francisco@ime.uerj.br`

Exercício (aula passada)

- movimentar dois pixels ao mesmo tempo
- usar um conjunto de teclas diferentes
 - como se fossem dois jogadores

Solução (aula passada)

```
emit FRAMES_SET(yes);
```

```
code/await Player(var int up,  
                  var int right,  
                  var int down,  
                  var int left,  
                  var Color color) -> NEVER do
```

```
//...
```

```
end
```

```
spawn Player(KEY_UP, KEY_RIGHT, KEY_DOWN, KEY_LEFT, COLOR_BLUE);
```

```
spawn Player(KEY_w , KEY_d      , KEY_s      , KEY_a      , COLOR_RED);
```

```
await FOREVER;
```

```
code/await Player(var int up,  var int right, var int down,
                  var int left, var Color color) -> NEVER do
var Point pt = val Point(0,0);
par do
  var int key;
  every key in KEY_PRESS do
    if (key == left) and (pt.x > -25) then
      pt.x = pt.x - 1;
    else/if (key == right) and (pt.x < 25) then
      pt.x = pt.x + 1;
    else/if (key == up) and (pt.y < 25) then
      pt.y = pt.y + 1;
    else/if (key == down) and (pt.y > -25) then
      pt.y = pt.y - 1;
    end
  end
end
with
  //every FRAME_REDRAW
end
end
```

```
code/await Player(var int up,  var int right, var int down,
                  var int left, var Color color) -> NEVER do

var Point pt = val Point(0,0);

par do

    //every key in KEY_PRESS do

with

    every FRAMES_REDRAW do
        emit GRAPHICS_SET_COLOR_NAME(color);
        emit GRAPHICS_DRAW_PIXEL(pt.x, pt.y);
    end
end

end

end
```

Pergunta

- Qual o problema dessa solução?
 - Tente movimentar os dois pixels ao mesmo tempo

Por que isso acontece?

- Os eventos do teclado só ocorrem quando um estado de uma tecla muda de pressionado para não pressionado e vice-versa
- É assim que o sistema operacional envia mensagens de uma maneira geral

Por que isso acontece?


- Mas por que conseguimos mover um pixel mantendo uma tecla pressionada?

Eventos periódicos

- O sistema operacional pode enviar eventos periódicos
 - se estiver configurado para isso

referente ao tempo para começar a repetição

referente ao tempo para repetição de cada tecla



“Exercício”

- Nas propriedades de teclado do seu sistema operacional, mude o **intervalo de repetição** e a **taxa de repetição** e veja o comportamento do exemplo **3-e-move2.ceu**

Como resolver?

- Não vamos mais tratar pressionamentos de teclas como um evento contínuo
- Vamos armazenar as teclas pressionadas em variáveis de estado

Solução

```
emit FRAMES_SET(yes);
```

```
code/await Player(var int up,  
                  var int right,  
                  var int down,  
                  var int left,  
                  var Color color) -> NEVER do
```

```
    //... (próximo slide)
```

```
end
```

```
spawn Player(KEY_UP, KEY_RIGHT, KEY_DOWN, KEY_LEFT, COLOR_BLUE);
```

```
spawn Player(KEY_w , KEY_d      , KEY_s      , KEY_a      , COLOR_RED);
```

```
await FOREVER;
```

```
var Point pt = val Point(0,0);
var int x_axis = 0; var int y_axis = 0;
par do
  var int key;
  every key in KEY_PRESS do
    //atualiza x_axis e y_axis com 1 ou -1
  end
with
  var int key;
  every key in KEY_UNPRESS do
    //atualiza x_axis e y_axis com 0
  end
with
  every FRAMES_UPDATE do
    //atualiza pt com base em x_axis e y_axis
  end
with
  every FRAMES_REDRAW do
    //desenha pixel
  end
end
```

```

var Point pt = val Point(0,0);
var int x_axis = 0; var int y_axis = 0;
par do
  var int key;
  every key in KEY_PRESS do
    //atualiza x_axis e y_axis com 1
  end
with
  var int key;
  every key in KEY_UNPRESS do
    //atualiza x_axis e y_axis com 0
  end
with
  every FRAMES_UPDATE do
    //atualiza pt com base em x_axis e y_axis
  end
with
  every FRAMES_REDRAW do
    //desenha pixel
  end
end
end

```

x_axis = 0

Não está se movendo horizontalmente

x_axis = -1

Está se movendo para esquerda

x_axis = 1

Está se movendo para direita

Exercício

- Implemente a nova versão do `code/await Player` usando a estrutura do slide anterior
 - disponível no arquivo **1-move2.ceu**
- Pense em quando `x_axis` e `y_axis` devem mudar de estado

Solução - KEY_PRESS

```
var int key;  
every key in KEY_PRESS do  
    //atualiza x_axis e y_axis com 1 ou -1  
    if (key == left) then  
        x_axis = -1;  
    else/if (key == right) then  
        x_axis = 1;  
    else/if (key == up) then  
        y_axis = 1;  
    else/if (key == down) then  
        y_axis = -1;  
    end  
end
```


Solução - KEY_UNPRESS

```
var int key;  
every key in KEY_UNPRESS do  
    //atualiza x_axis e y_axis com 0  
    if (key == left) or (key == right) then  
        x_axis = 0;  
    else/if (key == up) or (key == down) then  
        y_axis = 0;  
    end  
end
```

Solução - FRAMES_UPDATE

```
every FRAMES_UPDATE do
    //atualiza pt com base em x_axis e y_axis
    if (x_axis == 1) and (pt.x < 25) then
        pt.x = pt.x + 1;
    else/if (x_axis == -1) and (pt.x > -25) then
        pt.x = pt.x - 1;
    end

    if (y_axis == 1) and (pt.y < 25) then
        pt.y = pt.y + 1;
    else/if (y_axis == -1) and (pt.y > -25) then
        pt.y = pt.y - 1;
    end
end
```

end

Solução - FRAMES_REDRAW

```
every FRAMES_REDRAW do
    //desenha pixel
    emit GRAPHICS_SET_COLOR_NAME(color);
    emit GRAPHICS_DRAW_PIXEL(pt.x, pt.y);
end
```

2-e-move2.ceu

Exercício

- Como podemos mudar a velocidade do pixel?

Solução

- Substituir o **every FRAMES_UPDATE** do por **every 150ms** do

```
every 150ms do
  //atualiza pt com base em x_axis e y_axis
  if (x_axis == 1) and (pt.x < 25) then
    pt.x = pt.x + 1;
  else/if (x_axis == -1) and (pt.x > -25) then
    pt.x = pt.x - 1;
  end

  if (y_axis == 1) and (pt.y < 25) then
    pt.y = pt.y + 1;
  else/if (y_axis == -1) and (pt.y > -25) then
    pt.y = pt.y - 1;
  end
end
end
```

Desafio

- O que acontece quando fazemos a seguinte sequência?
 - clica na tecla para esquerda e a mantém pressionada
 - clica na tecla para direita e a mantém pressionada
 - solta a tecla da esquerda
- Isso se limita ao movimento horizontal (eixo-x)?
- Esse era o comportamento desejado?
- **Desafio:** Como resolver?